Share some of the mathematics you love with the readers of your web pages! Maple 8 makes it easy to create mathematical animations, and those animations can be saved as animated GIF files that will make your web pages come to life! Of course, an animation doesn’t show up very well on paper—you will have to visit our web page at http://www.jcu.edu/math/ICTCM2002/ to see the animations described here in action. Or—better yet—create some of your own using the techniques below.

Summary of Techniques:

1. **Create an animation with Maple 8.** Use the code on the following pages, or create your own.

2. **While interacting with Maple, reduce the size of the image.** Grab one of the corner handles and drag. Remember that your image has multiple frames, so a large image will result in a very large file size—especially with 3D graphs.

3. **While you’re interacting with Maple, adjust the color of your animation.** For a rotating 3D image, choose a color scheme in which the color of a point depends only on its z-coordinate. (We like to use Z(Hue).) This is especially important if you are taking advantage of symmetry in your image to make a 90 or 180 degree rotation seem like 360 degrees.

4. **After you are satisfied with the appearance of your animation, right click on the image and choose Export As ... GIF.** This process may take up to a minute or more, depending on how complex your animation is, and depending on the speed of your computer. Alternately, choose Export As ... HTML from the “file” menu. In this case, Maple will save an HTML page that contains your Maple code and the image, stored as an animated GIF file. The Windows version of Maple saves the GIF in a subfolder titled Images.

5. **Use an animation editor to do any additional editing to your image.** Some of the touch-up editing that you may wish to do to your image:
• **Slow it down!** With today's computers getting faster and faster, your animation may be moving at warp speed on some viewers’ machines.

• **Crop out extra white space.** Images directly from Maple are frequently surrounded by a great deal of empty space that can make it impossible to position your animation properly on a web page.

• **Add transparency to the GIF or change the background color.** This is important if you like to write web pages with backgrounds other than plain white.

• **Make it pause at the end**, before beginning the animation cycle again. For example, in our trefoil knot animation we pause after the knot is completely tied, before tying it all over again.

There are a number of animation editors on the market. We like to use Animation Shop, which comes with Paint Shop Pro. (See http://www.jasc.com.)

6. **Include your animation on your web page.** All of the commonly used HTML editors include a way for you to insert images in web pages. Since animations created in this way are simply animated GIF files, no special handling is needed to insert the image.

7. **Enjoy your new web page!!**

Some Comments:

Although we used Maple 8 to create these animations, all of the Maple code in this paper, and most of the Maple code on our web site, works in earlier versions of the software, down to Maple V Release 5.

In spite of the fact that our primary motivation was simply to have some mathematical fun, some students have observed that several of these animations can be instructive as well! In particular, the slope field and Riemann sum animations included in this paper illustrate some important mathematical ideas.

Conclusion:

Inclusion of mathematical animations on your web pages can help bring life to otherwise static pages. However, we caution you not to over-use these techniques, as we firmly believe that the effectiveness of these animations is greatly reduced by using them too frequently. Too many spinning, flapping or twisting images can be distracting to the readers of your web pages.

The authors can be contacted by e-mail at bdambrosia@jcu.edu and spitz@jcu.edu. Live examples of the animations discussed in this paper can be found on our web page, at http://www.jcu.edu/math/ICTCM2002/
Rotating 3D Surface

Maple 8 code:

```maple
with(plots):
pic := n ->
    [plot3d(sin(sqrt(x^2+y^2)),x=-5..5,y=-5..5,
        orientation=[3*n,67],grid=[20,20])]:
display(seq(pic(n),n=0..29),insequence=true);
```

Comments:

- It is actually the viewpoint that is being rotated by this Maple code, rather than the surface. However, the effect is that of the surface rotating in 3-space.
- The 4-way symmetry of this surface makes it possible to get by with rotating through only 90 degrees, rather than 360. This results in a considerably smaller animated GIF file. If you use a surface with less symmetry, you will have to rotate through a larger angle, resulting in more frames in the animation and a larger file size.
- Choose a color scheme in which the color of a point is determined by its z-coordinate; otherwise, there may be an abrupt color change when the rotation passes 90 degrees and starts again at 0 degrees. (We like to use Z(Hue).)

Morphing Limaçons

Maple 8 code:

```maple
with(plots):
animate(1+(2-4*abs(x))*cos(theta), theta=0..2*Pi,
    x=-1..1,coords=polar,thickness=2,numpoints=100,
    scaling=constrained,tickmarks=[0,0],frames=50,
    labels=[``,``],color=blue);
```

Comments:

- The family of limaçons \( r = 1 + a \cos(\theta) \) is varied by making \( a \) range from -2 to 2, and then back to -2 again. This is done by the function \( a = 2 - 4|x| \), with \(-1 \leq x \leq 1\).
- An animation editor was used to slow down the animation. Each frame is displayed for 0.1 seconds, except the end frames, which were given longer display times.
**Tying a Trefoil Knot**

**Maple 8 code:**

```maple
with(plots):
TieKnot := proc(n :: posint)
    local i, t, curve, pict;
    curve := [-10*cos(t) - 2*cos(5*t) + 15*sin(2*t),
              -15*cos(2*t) + 10*sin(t) - 2*sin(5*t),
              10*cos(3*t)];
    pict := [seq(tubeplot(curve, t=0..2*Pi*i/n,
                           radius=3), i=1..n)];
    display(pict, insequence=true);
end;
TieKnot(48);
```

**Comments:**

- The image above was created with 48 frames. You can experiment with other numbers of frames, as well as with various parametric curves and various values for the radius of the tube.
- An animation editor was used to slow down the display of the frames, as well as to add a 3-second display time to the final frame.
- This code was taken from p. 276 of the Maple 8 Advanced Programming Guide.

**Trefoil Knot with Traveling Bead**

**Maple 8 Code:**

```maple
with(plots): with(plottools):
P:=tubeplot([(2+cos(1.5*t))*cos(t),
              (2+cos(1.5*t))*sin(t),sin(1.5*t)],
            t=0..4*Pi,radius=0.4,shading=XY):
pic:=n->sphere([(2+cos(1.5*n*Pi/6))*cos(n*Pi/6),
                 (2+cos(1.5*n*Pi/6))*sin(n*Pi/6),
                 sin(1.5*n*Pi/6)],0.6,color=red):
Q:=display(seq(pic(s),s=0..23),insequence=true,
           scaling=constrained):
display({P,Q});
```

**Comments:**

- The bead shape could also be any of the polyhedra types supported by Maple.
- An animation editor was used to slow down the display time of each frame, and to crop out the excess white space around the animation.
Slope field and solutions to a differential equation

Maple 8 code:

```maple
with(DEtools): with(plots):
P:=dfieldplot(diff(y(x),x)=y(x)+x^2,y(x),
   x=-2..2,y=-2..2,thickness=1):
f:=x->-x^2*exp(-x)-2*x*exp(-x)-2*exp(-x)+1;
g:=x->exp(x)*f(x);
g1:=x->exp(x)*(f(x)+1);
Q1:=animatecurve(g(x), x=-2..2,
    color=black,thickness=2,
    frames=50, xtickmarks=0, ytickmarks=0):
R:=plot(g(x),x=-2..2, color=black,
    thickness=2,xtickmarks=0,
ytickmarks=0):
Q2:=animatecurve(g1(x), x=-2..2,
    color=blue, thickness=2,
    frames=50, xtickmarks=0, ytickmarks=0):
display({P,Q1}); display({P,R,Q2});
```

Comments:

- This is actually two animations that we concatenated in our animation editor.
- We lengthened the display time for every frame, and further lengthened the time for the first and last frames, as well as the frame that shows the completed first curve.

Riemann Sums Converging to an Integral

Maple 8 code:

```maple
with(plots): with(student):
pic := n->
    [rightbox(x+sin(x)+3,x=0..2*Pi,2*n,
tickmarks=[0,0])]:
display(seq(pic(n),n=1..20),insequence=true);
```

Comments:

- This animation shows Riemann sums for a function, using \( n = 2, 4, 6, 8, \ldots, 40 \) subintervals.
- It will have to be slowed down for display on almost all computers. Our version displays frames for varying times, the first and last frames being displayed for the longest times.